

# Design of loaded and trap dipoles — using a computer program

This article describes a method of designing simple HF antennas which are self-resonant on one or more frequencies. It includes a computer program (in BASIC) to simplify the calculations involved.

A. P. Boon VK7AW

A SUITABLE ANTENNA is a very necessary companion for any radio transmitter. There are many different types to choose from — multiband, broadband, monoband — to name but a few. An antenna can be self-resonant on the operating frequency or can be teamed with an antenna tuning unit which transforms the antenna impedance to the impedance required by the transmitter (usually 50 ohms resistive). This article describes a method of designing antennae which are self-resonant on one or more frequencies, and includes a computer program to simplify the calculations involved.

## The loaded dipole

A dipole antenna which is shorter than a half wavelength can be resonated at a given frequency by adding lumped inductance as shown in Figure 1. Provided that the value of  $L$  is chosen correctly, this antenna will behave like a physical halfwave dipole, at the resonant frequency. The direction of maximum radiation is at right angles to the dipole, and the impedance is resistive and about 70 ohms (depending on the height above ground), although the bandwidth for a given VSWR is less than for a full length dipole. Such a configuration often has advantages, especially in restricted-space applications.

For the person who wishes to build such a dipole with a minimum of fuss, it is useful to be able to calculate the values of  $L$ , the loading inductance, before constructing the prototype. In an article in *QST*, September 1974, Jerry Hall (K1PLP) published an empirical formula giving the value of  $L$ , shown in the panel here.




Figure 1. Loaded dipole.

This is a bit of a mouthful, even if you have a scientific programmable calculator! Subsequently, an article by Dick Sander (K5OY) appeared in *CQ*, December 1981, with a computer program for calculating the value of  $L$  for a simple loaded dipole. The program applied the formula directly, substituting wire gauge for diameter.

The main routine of the program described here uses the formula of K1PLP, rearranged for metric dimensions. Additional routines are included to ensure that most of the facilities required for the design of loaded, trapped and loaded-trap dipoles are available in the one program.

## Program description

The program is written in BASIC and provides eight options which appear in a menu (lines 200-285). After an option is selected and executed, the user is given the choice of performing further calculations (line 340), returning to the menu if so desired. Some checking of the range of input data is included, but gross errors (e.g: trying to input alphabetical characters in response to a request for a numerical value) will cause the program to halt abruptly.

The first option available calculates the loading inductance required for a dipole

which is shorter than a half wavelength at its operating frequency. The input data are *frequency* (MHz), *overall length* (metres), the *distance of the loading coil* from the centre of the dipole (metres) and the *dipole wire or tube diameter* (mm).

Intermediate expressions S1 to S5 are used to make the final expression for  $L$  (represented by S6 in line 1280) less complex than the original formula.

The second option on the menu calculates the physical length necessary to make a dipole resonant on a given frequency, with loading coils of value  $L$  micro-henries at a distance of  $B$  metres from the centre. This, in a sense, is the inverse of the first option, and uses an interactive technique to find the resonant length. The assumption is made that the resonant length lies outside the loading coils, but inside a half wavelength.

The initial guess is the average of these two values, and part of option 1 is called as a subroutine to calculate the value of loading required if the antenna were that length. The calculated value of loading is compared with the actual value, and the decision is made as to whether the resonant length is longer or shorter than the initial guess. This process is then repeated, with the interval of uncertainty being halved at every iteration.

The process halts when the calculated inductance for an estimate is within 10 mH of the actual value. Such a scheme may not be an optimum strategy for this particular calculation. However, it is only carried out once, and therefore the processing time is not significant.

Option 3 is a straightforward conversion of inductance into inductive reactance, at a given frequency. The inclusion of this function makes life a little easier in the design of trap dipoles, as will be seen later.

Option 4 calculates the values of inductance and capacitance required for a parallel LC trap, given the resonant frequency and the inductive reactance at a lower frequency. This function is essential for the design of trap dipoles.

Option 5 also calculates values for an LC trap. Given any two of the resonant frequency, the inductance and the capacitance, it will calculate the third value. ▶

$$L = \frac{10^6}{68\pi^2 f^2} \left\{ \left[ \ln \left( \frac{24 \left( \frac{234}{f} - B \right)}{D} \right) - 1 \right] \left[ \left( 1 - \frac{fB}{234} \right)^2 - 1 \right] - \left[ \ln \left( \frac{24 \left( \frac{A}{2} - B \right)}{D} \right) - 1 \right] \left[ \left( \frac{fA}{2} - fB \right)^2 - 1 \right] \right\} \mu\text{H}$$

where  $A$  is the overall length of the dipole in feet,  $B$  is the distance from the centre of the dipole to the loading coil.  $D$  is the diameter of the radiator (wire or tube) in inches, and  $f$  is the frequency in MHz.

Option 6 calculates the inductive reactance of an LC trap at a frequency below the resonant frequency. The inductive reactance is then converted into an equivalent inductance at the lower frequency; this calculation is also required in the design of trap dipoles.

Option 7 applies the well known formula (found in any ARRL *Radio Amateurs Handbook*) to find the number of turns required on a former to provide a specified inductance value. Option 8 is the way out.

## APPLICATIONS

### Simple loaded dipole

Determining the loading inductance required to resonate a dipole with an overall length less than a physical half wavelength is a straightforward application of menu option 1. For example, if you have space for a dipole 20 metres long, and you wish to load it for operation on 3.6 MHz, then select option 1 from the menu, enter 3.6 for the frequency, 20 metres for the overall length, your choice of the distance of the loading coil from the centre of the dipole, and the diameter of the wire being used.

The distance of the coil from the centre is not predetermined — you must specify it. The further from the centre is the coil, the higher the value of L, and thus the higher the resistance of the coil, for a given wire diameter. On the other hand, the closer the coil is to the centre, the higher will be the current flowing in it, and thus the higher the power dissipated in the (lower) resistance of the coil.

There will be an optimum position for maximum efficiency for a given set of physical parameters (dipole length, wire size, etc.), but this position is not determined by the program. Thus, there remains some scope for experimentation!

To get back to the example, if you select a position 5 metres from the centre and use wire 1.6 mm in diameter, the program will respond with "Loading Coil is 41.6 microhenry", and give you the option of doing further calculations. You can then select menu item 7 to calculate the number of turns required with a given former and wire size, to produce the required inductance.

### Two-frequency trap dipole

Let us suppose you have space for a dipole 30 metres long, and you wish to use it as 3.6 and 7.05 MHz. Since 30 metres is longer than a half wavelength at 7.05 MHz you can put traps, resonant at 7.05 MHz, a quarter wavelength from the centre, thus achieving a half-wave dipole at 7.05 MHz; the trap is inductive at 3.6 MHz, and will act as a loading coil at 3.6 MHz. The requirement is to select the inductance (and thence the capacitance) of the trap correctly. Proceed as follows:

- (i) Select menu option 1.
- (ii) The operating frequency is 3.6 MHz.
- (iii) The overall length of the dipole is 30 metres.
- (iv) Calculate a quarter wavelength at

7.05 MHz, less 5% to be 10.11 metres (not in the program — use your calculator!). This is the distance of the loading coil (i.e. the trap) from the centre of the dipole.

- (v) Input the wire diameter, say 2 mm.
- (vi) The program gives the required loading, 28.0 microhenry.
- (vii) Use menu option 3 to convert the loading inductance (28.0  $\mu$ H) into inductive reactance at 3.6 MHz. The program returns a value of 633.35 ohms.
- (viii) Use menu option 4 to calculate the trap values required. The resonant fre-

quency is 7.05 MHz, the trap has an inductive reactance of 633.35 ohms at 3.6 MHz. The program returns values of 20.7 microhenry and 24.6 picofarad.

By using parallel resonant traps of 20.7  $\mu$ H and 24.6 pF, the dipole will be resonant on both 3.6 and 7.05 MHz.

Of course, the value of 24.6 pF is not a preferred value. To use the nearest standard value (27 pF), calculate the inductance necessary to resonate with 27 pF at 7.05 MHz, using menu option 5. Enter "?" for the value of inductance, and the program will calculate the required value, in

```

100 REM * LOADED DIPOLE DESIGN PROGRAM.
120 REM * WITH ADDITIONAL ROUTINES NECESSARY FOR TRAP AND
130 REM * LOADED-TRAP DIPOLE DESIGN.
140 REM *
160 REM * PREPARED BY ANDREW BOON, VK7AW, 12th OCTOBER 83.
170 REM *
180 REM * Menu of available functions.
200 PRINT: PRINT "The Following Functions Are Available:": PRINT
230 PRINT "1. Calculate Inductance of Loading Coil for Loaded Dipole."
235 PRINT "2. Calculate Length of Dipole with Given Loading Inductances."
240 PRINT "3. Convert Inductance into Inductive Reactance."
250 PRINT "4. Calculate L,C for a Trap Filter, Given f, fL & XL."
260 PRINT "5. Calculate f, L, or C for a Trap (one unknown)."
270 PRINT "6. Calculate Equivalent Inductance of a Trap at Lower Freq."
280 PRINT "7. Calculate Number of Turns For a Given Inductance."
285 PRINT "8. Exit From This Program.": PRINT
300 INPUT "Your Selection (1 to 8, Return) ":S
310 IF (S<>INT(S)) OR (S<1) OR (S>8) THEN 300
315 IF S=8 THEN END
320 ON S GOSUB 1000,2000,3000,4000,5000,6000,7000
340 PRINT: INPUT "Any More Calculations (Y/N, Return) ":A$
350 IF (A$<>"Y") AND (A$<>"N") THEN 340
360 IF A$="Y" THEN 200 ELSE END
1000 REM * Dipole Loading Coil Calculation.
1010 REM * Based on formula of Jerry Hall (K1PLP) in "QST", Sept. 1974.
1020 REM *****
1040 PRINT: PRINT "Dipole Loading Coil Calculation.": PRINT
1060 INPUT "Operating Frequency in MHz ":f
1065 IF f>0 THEN 1070: GOSUB 10000: GOTO 1060
1070 INPUT "Total Length of Dipole in Metres ":A
1075 IF A>0 THEN 1080: GOSUB 10000: GOTO 1070
1080 INPUT "Distance from Centre of Dipole to Loading Coil, metres ":B
1085 IF B>0 AND B<A/2 THEN 1090: GOSUB 10000: GOTO 1080
1090 INPUT "Diameter of Wire used in the Dipole (mm) ":D
1095 IF D>0 THEN 1098: GOSUB 10000: GOTO 1090
1098 GOSUB 1200
1100 PRINT: PRINT USING "Loading Coil is ###.#",S6:
1105 PRINT " microhenries.": PRINT
1110 RETURN
1200 REM Calculate loading coil inductance.
1210 REM For formula see "QST", Sept. 1974, p.28.
1230 S1=1E6/(223.1*(PI*f)^2): S2=71.32/f-B
1250 S3=1-f*B/71.32: S4=A/2-B: S5=2000/D
1280 S6=S1*((LOG(S5*S2)-1)*(S3^2-1)/S2 -(LOG(S5*S4)-1)*((S4*f/71.32)^2-1)/S4)
1290 RETURN
2000 REM * Determine Physical Length of a Resonant Dipole with Loading Coils
2020 REM * of Inductance L microhenry, B metres from Centre of Dipole.
2040 REM *****
2050 INPUT "Value of Loading Coils, uH ":L
2055 IF L>0 THEN 2060: GOSUB 10000: GOTO 2050
2060 INPUT "Frequency of Operation (MHz) ":f
2065 IF f>0 THEN 2070: GOSUB 10000: GOTO 2060
2070 INPUT "Diameter of Dipole Wire (mm) ":D
2075 IF D>0 THEN 2080: GOSUB 10000: GOTO 2070
2080 INPUT "Distance of Loading Coil from centre (m) ":B
2082 REM Check that Coils are not too far out.
2090 W=0.5*300/f+0.95: REM 95% of a half-wavelength.
2100 IF (2*B<W) THEN 2120
2110 PRINT "Loading Coils Beyond Quarter-Wavelength.":RETURN
2120 MI=2*B: MA=W: REM Minimum and Maximum possibilities.
2125 A=0.5*(MA+MI): REM Next Guess for A is midway between Min and Max.
2130 GOSUB 1200: REM Calculate Loading Coil for this Position.
2140 IF ABS(L-S6)<0.01 THEN GOTO 2170
2150 IF S6<L THEN MA=A ELSE MI=A
2160 GOTO 2125
2170 PRINT USING "Resonant Length with Loading Coil is ####.##",A:
2175 PRINT " metres.": PRINT
2190 RETURN
3000 REM * Convert Inductance to Inductive Reactance, at a given Frequency.
3020 REM *****
3030 INPUT "Inductance, microhenry ":L
3035 IF L>0 THEN 3040: GOSUB 10000: GOTO 3030
3040 INPUT "Frequency, MHz ":f
3045 IF f>0 THEN 3050: GOSUB 10000: GOTO 3040
3050 PRINT USING "Inductive Reactance is ####.##",2*PI*f*L:
3060 PRINT " Ohms.": PRINT

```

# Design of loaded and trap dipoles

this case 18.9 microhenry. Then, using menu option 6, the equivalent inductance of the trap at 3.6 MHz is calculated i.e.: 25.6  $\mu$ H. Use menu option 2 to calculate the exact length of a loaded dipole operating at 3.6 MHz with a 25.6  $\mu$ H loading coil at a distance of 10.11 metres from the centre. The value obtained is 30.48 metres. By cutting the dipole to 30.48 metres overall length (including the physical length of the loading coil formers), you have a dipole which is resonant on both 3.6 and 7.05 MHz, and a trap which uses standard values of capacitance, i.e: 27 pF.




Figure 2. Loaded trap dipole.

## Loaded trap dipole

For applications in more restricted space than Example 2, the dipole could be loaded for operation on the higher frequency. For example, if only 20 metres was available for the dipole and operation on 3.6 and 7.05 MHz was required, one possibility

would be to place a loading coil three metres from the centre, and a 7.05 MHz trap six metres from the centre. We thus form a loaded dipole on both frequencies, with two pairs of loading coils for the lower frequency, as in Figure 2. Proceed as follows:

(i) Select option 1 to calculate the value of the inner loading coils, L1. Use 7.05 MHz, 12 metres as the overall length, three metres from the centre to the loading coil and, say, 2 mm wire. The program calculates 14.4  $\mu$ H as the loading coil value.

(ii) Use option 2 to calculate the physical length of a dipole resonant at 3.6 MHz, when loading coils of 14.4  $\mu$ H are placed three metres from the centre. The value calculated is 27.21 metres. The full span of a half wave dipole at 3.6 MHz is 95% of the actual value, i.e: 39.58 metres. Thus the loading coil has the effect of shortening the half wave dipole by 39.58 - 27.21 = 12.37 metres.

(iii) Use option 1 to calculate the inductance required of the traps to resonate a dipole of length 20 + 12.37 = 32.37 metres, at 3.6 MHz, when the trap (loading coil) is placed 6 + 12.37/2 = 12.19 metres from the centre. This gives a value of 30.3  $\mu$ H.

(iv) Use option 3 to convert 30.3  $\mu$ H to inductive reactance at 3.6 MHz. The value returned is 685.37 ohms.

(v) The trap must resonate at 7.05 MHz and have an inductive reactance of 685.37 Ohms at 3.6 MHz. Use option 4 to calculate the values of L2 and C for the trap. The program returns values of 22.4  $\mu$ H and 22.8 pF.

Once again, a similar procedure can be followed as for Example 2, to calculate the value of L2 and the length of the outer section of the dipole, if a standard value of capacitance (e.g: 22 pF) is used.

No more examples will be presented here, but the procedures detailed above can be extended, for example, to tri-band trap dipoles. Simply work outwards from the centre, considering the highest frequency first; the trap for the highest frequency becomes a loading inductance for the lower frequencies. The middle frequency uses a loaded trap dipole (as for Example 3), and the lowest frequency has two loading inductances.

As regards the accuracy of the calculations, K1PLP made the comment (in the QST article) that "the final inductance values found by cut-and-try pruning for lowest SWR at the desired frequency have been so close to the value from calculations that a laboratory bridge was necessary to measure the difference." This was referring to simple loaded dipoles.

A loaded trap dipole was recently designed using the procedure of Example 3; the traps and loading coils were made up to the calculated values, and when the antenna was erected, no pruning was required. The SWR minima occurred within 3 kHz of the design values.

```

3070 RETURN
4000 REM * Calculate Values of L,C for a Trap Filter, given the Resonant
4010 REM * Frequency and the Inductive Reactance at a Lower Frequency.
4020 REM *****
4030 INPUT "Resonant Frequency of Trap, MHz " :f0
4040 IF f0>0 THEN 4050: GOSUB 10000: GOTO 4030
4050 INPUT "Inductive Reactance at Lower Frequency, Ohms " :XL
4060 IF XL>0 THEN 4070: GOSUB 10000: GOTO 4050
4070 INPUT "Lower Frequency, MHz " :fL
4080 IF fL>0 AND fL<f0 THEN 4090: GOSUB 10000: GOTO 4070
4090 W0=2*PI*f0: WL=2*PI*fL
4100 C=WL/(XL*(W0*W0-WL*WL))*1E6
4110 L=1/(W0*W0*C)*1E6
4120 PRINT USING "Values for Trap are: L=####.#,L: PRINT " uH"
4130 PRINT USING " C=####.#,C: PRINT " pF"
4140 RETURN
5000 REM * Find Resonant Frequency, Inductance or Capacitance of a Parallel
5010 REM * Tuned Circuit, given 2 out of 3 of the values.
5020 REM *****
5030 INPUT "Resonant Frequency, MHz (or '?' if unknown) " :f0$
5040 IF f0$="" OR VAL(f0$)>0 THEN 5050: GOSUB 10000: GOTO 5030
5050 INPUT "Inductance, microhenry (or '?' if unknown) " :L$
5060 IF L$="" OR VAL(L$)>0 THEN 5070: GOSUB 10000: GOTO 5050
5070 INPUT "Capacitance, picofarad (or '?' if unknown) " :C$
5080 IF C$="" OR VAL(C$)>0 THEN 5090: GOSUB 10000: GOTO 5070
5090 IF f0$="" AND VAL(L$)>0 AND VAL(C$)>0 THEN 5200
5100 IF L$="" AND VAL(f0$)>0 AND VAL(C$)>0 THEN 5300
5110 IF C$="" AND VAL(f0$)>0 AND VAL(L$)>0 THEN 5400
5120 PRINT "Must have 2 values given and 1 unknown.": GOTO 5030
5200 REM Calculate Resonant Frequency.
5210 L=VAL(L$): C=VAL(C$): f0=1E3/(2*PI*SQR(L*C))
5220 PRINT USING "Resonant Frequency is ####.#,f0: PRINT " MHz."
5230 RETURN
5300 REM Calculate Inductance.
5310 f0=VAL(f0$): C=VAL(C$): L=1E6/(4*PI*PI*f0*f0*C)
5320 PRINT USING "Inductance Required is ####.#,L:PRINT " microhenry."
5330 RETURN
5400 REM Calculate Capacitance.
5410 f0=VAL(f0$): L=VAL(L$): C=1E6/(4*PI*PI*f0*f0*L)
5420 PRINT USING "Capacitance Required is ####.#,C: PRINT " pF."
5430 RETURN
6000 REM * Calculate the Equivalent Inductance of a Parallel Resonant
6010 REM * Trap, at a Frequency below the Resonant Frequency.
6030 REM *****
6100 INPUT "Resonant frequency of Trap, MHz " :f0
6110 IF f0>0 THEN 6120: GOSUB 10000: GOTO 6100
6120 INPUT "Inductance of Trap, microhenry " :L
6130 IF L>0 THEN 6140: GOSUB 10000: GOTO 6120
6140 W0=2*PI*f0: C=1E6/(W0*W0*L)
6150 PRINT USING "(Capacitance should be ####.#,C: PRINT " pF.)"
6160 INPUT "Lower Frequency, MHz " :fL
6170 IF fL>0 AND fL<f0 THEN 6180: GOSUB 10000: GOTO 6160
6180 WL=2*PI*fL: LE=L/(1-(WL*WL)/(W0*W0))
6190 PRINT USING "Equivalent Inductance at Lower Frequency is ####.#,LE:
6200 PRINT " microhenry."
6210 RETURN
7000 REM * Calculate Number of Turns For a Given Inductance Value.
7020 REM * Coil Former Diameter, Pitch and Wire Diameter.
7030 REM *****
7040 INPUT "Diameter of Coil Former (mm) " :A
7050 IF A>0 THEN 7060: GOSUB 10000: GOTO 7040
7060 INPUT "Diameter of Wire (mm) " :D
7070 IF D>0 AND D<A/2 THEN 7080: GOSUB 10000: GOTO 7060
7080 INPUT "Coil Pitch (mm per turn) " :P
7090 IF P>D THEN 7100: GOSUB 10000: GOTO 7080
7100 INPUT "Inductance Required (uH) " :L
7110 IF L>0 THEN 7120: GOSUB 10000: GOTO 7100
7120 A=A+D: REM Calculate average diameter of coil.
7125 B=1016*P*L/(A*A)
7130 N=(B + SQR(B*B + 1828*L/A))/2
7140 REM Formula from ARRL Handbook, re-arranged and metricated.
7150 PRINT USING "Number of Turns Required is ####.#,N: PRINT
7170 RETURN
10000 REM * Subroutine executed when out-of-range data is detected.
10020 REM *****
10030 PRINT "Check the Value of Your Data Input.": PRINT
10050 RETURN
    
```